
Abstract

This paper presents a new approach for the formal specification of cryptographic schemes using
types. It discusses specifying a cryptographic protocol using homotopy type theory which adds the
notion of higher inductive type and univalence to Martin-Löf’s intensional type theory. A higher in-
ductive type allows us to introduce constructors for paths and higher-dimensional paths in addition to
points. Equivalences or bijections in the universe can identify the paths through univalence. A higher
inductive type specification can act as a front-end mapped to a concrete cryptographic implementa-
tion in the universe. By having a higher inductive type front-end, we can encode domain-specific
laws of the cryptographic implementation as higher-dimensional paths. Due to univalence and func-
toriality of mappings in homotopy type theory, the path structure will be preserved in the mapping
and realized by equivalence in the universe. The higher inductive type gives us a graphical compu-
tational model and can be used to extract functions from underlying concrete implementation. Using
this model we can achieve various guarantees on the correctness of the cryptographic implementation.

Keywords: Higher Inductive Type, Univalence, Functor, Homotopy, Groupoid, Universe, Equiva-
lence, Quasi-inverse, Identity type

1 Introduction

Formal verification of cryptographic protocols has become a significant research focus over recent years
[27] [9]. Some widely used cryptographic implementations were found to be flawed after their deploy-
ment becoming vulnerable to various attacks. For example, the Heartbleed attack (CVE- 2014-0160)
is a consequence of a simple coding error [17]. Even with skilled designers, developers and testers it
is highly difficult to implement a cryptographic protocol without errors [20]. Theorem provers with a
mathematical background such as Coq [31] and Agda [25] can be used for implementing cryptographic
protocols and for designing attack models [27].

Formal methods are used to verify that a system behaves in an expected way based on its specifi-
cation. Type system, a lightweight formal method, is a tool for reasoning about programs which can
categorically prove the absence of some bad program behaviors. In the early days of programming, type
systems were used to ensure certain basic correctness properties of programs such as the arguments to
primitive arithmetic operations are always numbers and differentiating between a string and integer value
in the memory. During the twentieth century, types have become standard tools in logic, particularly in
proof theory. One of the significant work in this area is a predicative modification of Church’s type
system proposed by Per Martin-Löf now known as Martin-Löf type theory [23][22]. It gives a compu-
tational interpretation to intuitionistic higher-order logic based on Russell’s theory of types [30]. This
extended type systems from tools merely ensuring correctness properties into first-class logics.

Homotopy type theory [32] extends Martin-Löf intensional type theory by adding higher inductive
type and univalence axiom. In homotopy type theory, the witness or proof element of a type can be

viewed as a point in a topological space, and a witness of an identity type can be viewed as a path
in a topological space. A higher inductive type differs from an ordinary inductive type by providing
constructors not only for points but also for paths. In particular, higher inductive types provide a natural
encoding of many otherwise-difficult mathematical concepts, and univalence lets us work in our type
theory the way we do on paper: up to isomorphism. Homotopy type theory, however, is not yet done.
We do not yet have a mature theory or a mature implementation. While work proceeds on prototype
implementations of higher-dimensional type theories [4][14], much work remains before they will be as
convenient for experimentation with new ideas as Coq, Agda, or Idris is today. In the meantime, it is
useful to be able to experiment with ideas from higher-dimensional type theory in our existing systems.

Homotopy type theory has thus far primarily been applied to the encoding of mathematics, rather
than to programming. Nevertheless, some preliminary applications of homotopy type theory in program-
ming have been investigated. For example, the work of [21] apply ideas related to homotopy type theory
to modeling variable binding. Containers [3][1] in homotopy type theory can be used to implement data
structures such as multisets and cycles. Homotopical patch theory [5] shows modeling of Darcs [29]
version control system using the concepts of homotopy type theory. It models the patches as paths in a
higher inductive type. Application of logic to novel problems raises numerous interesting research issues
which could drive the progress in the theory. In this paper, we investigate a preliminary application of
homotopy type theory in cryptography and discuss its practical limitations from an application perspec-
tive. More specifically we discuss how to specify a cryptographic scheme, which is deterministic, using
the features of homotopy type theory.

We discuss specifying the correctness properties of a cryptographic implementation using higher in-
ductive types, implemented in Agda, and how to project computational models from such specifications.
A cryptographic system which expresses decryption as an inverse of encryption [16] can be defined using
a higher inductive type representing a graphical model in a topological space. A concrete implementa-
tion of the cryptographic system can be projected from this graphical model using univalence axiom.
The higher inductive type acts as an abstract model for the encoded cryptographic system and enables
us to specify the correctness properties as paths or higher-dimensional paths in a topological space. We
investigate the practical application of homotopy type theory to an industry level cryptographic protocol,
the cryptDB which employs multiple encryptions. We discuss the conceptual and the implementation
concerns and analyze the challenges from an application perspective discussing the limitations and the
future work. In short, we show how to extend types to act as formal certificates guaranteeing on various
correctness properties of a cryptographic scheme. Mainly we make the following contributions.

• We show how to design a cryptographic construction using a higher inductive type and how to
map the abstract type to a concrete implementation in the universe. Such developments give rise to
interesting homotopies which are paths between paths or two-dimensional paths in a topological
space.

• Paths in a higher inductive type are used to model correctness rules such as functional correctness
[16], which says decryption inverses encryption, and this structure will be preserved in the mapping
to the universe due to the functoriality of mappings in homotopy type theory.

• We can enforce various restrictions on the concrete implementation when a cryptographic protocol
is modeled using a higher inductive type. We discuss designing a higher inductive type for a
database model with multi-layered encryptions in the style of cryptDB [28].

• We discuss encoding of domain-specific properties related to homomorphic encryption, determin-
istic encryption, and order-preserving encryption as a path between paths or homotopies in a higher
inductive type.

We use the singleton framework [5] to implement the cryptDB protocol model discussed in this
paper. This framework can be generalized to support the class of protocols implemented with encryption
schemes that can be expressed using a contractible type. This limitation is imposed by homotopy type
theory which requires the paths to be bijective. However, we do not have a better framework yet to
implement non-bijective constructions. Designing cryptographic constructions as a higher inductive type
has the following benefits.

• In type theory all functions are functorial. Therefore, the functional correctness and domain-
specific properties of a cryptographic construction can be specified as paths or homotopies in a
higher inductive type, and the functions will preserve the path structures in the mapping of the
type to the universe.

• By specifying cryptographic properties as paths, we achieve guarantee on the correctness of the
underlying concrete implementation with respect to the encoded properties.

• We can have a graphical representation of a cryptographic construction in a topological space, and
we map it to a concrete implementation in the universe.

• By modeling a cryptographic construction as a higher inductive type, we can get the groupoid
structure and the relevant coherence laws related to the higher inductive type for free.

• We will get a non-dependent eliminator also known as the recursion principle, and we can use it to
define functions or to map the elements including the paths of the higher inductive type to elements
of other types such as the universe.

• We will get a dependent eliminator also known as induction principle which can be used to for-
mulate and prove theorems related to a cryptographic construction encoded as higher inductive
type.

• Correctness and theorem proving become an inherent part of the system. So we can eliminate the
cost involved in using an external theorem prover to reason about the correctness properties of the
cryptographic protocol in development.

In the next section, we will discuss the different components of homotopy type theory including
higher inductive type, univalence and functoriality. In section 3 we will give an example of encoding
a simple cryptographic scheme, the one-time pad, using higher inductive type and explain how to map
this higher inductive type to a concrete implementation of the scheme in the universe. In section 4, we
will discuss how to design higher dimensional paths to enforce restrictions on the implementation of
an industry level cryptographic protocol, the cryptDB. In section 5, we will review the implementation
details, and in section 6 we will discuss the related work. In section 7, we will see the limitations and
future work before concluding.

2 Background

A formal specification of a cryptographic scheme requires a programming language with support for
theorem proving. Proof-assistants with a strong mathematical background such as Agda and Coq can
be used to specify correctness and security properties of a cryptographic construction. There are works
which use an embedded domain-specific language [27] [9] [7] on existing theorem provers to support
defining and proving cryptographic properties. In this paper, we discuss a new approach to specify
cryptographic protocols based on types. This approach involves correlating a type with a cryptographic

implementation. By combining with the right type, we can guarantee on various correctness properties
of the cryptographic application. In the remainder of this section, we discuss the tools of homotopy type
theory which are instrumental in modeling and associating types with cryptographic implementations.

Unlike set theory, which is an interplay between propositions and sets, type theory is based on the
interpretation of propositions-as-types. According to this interpretation, a proposition stating that two
elements of a type a,b : A are equal corresponds to a type known as the identity type given by a =A b or
IdA(a,b). In homotopy type theory, elements of the identity type a =A b are used to model the notion
of paths or equivalences between a and b in the space A. An element of the type a =A b is a witness or
a proof stating that a and b are propositionally equal. Propositional equality is a proof relevant notion
of equality expressed by identity types. There is also a proof-irrelevant notion of equality in type theory
known as judgmental equality or definitional equality. Definitional equality is not internal to the theory,
and it is used to express equality by definition. For example, when we have a function f : Nat → Nat
defined as f (x) = x3 then f (2) is definitionally equal to 23.

Homotopy type theory extends Martin-Löf’s intensional type theory by adding univalence axiom
and higher inductive types. It introduces the notion of viewing type as a topological space in homotopy
theory or a higher-dimensional groupoid in category theory. Because of this correspondence, we can
observe an element of the identity type x =A y for a,b : A as a path in a topological space or a morphism
in a groupoid. Also, an element of the iterated identity types m =x=Ay n and p =m=x=Ayn q can be viewed
as a 2-dimensional and a 3-dimensional path respectively in a topological space or a morphism between
morphism and a higher-level morphism respectively in a groupoid and so on.

A morphism at a level k in a groupoid is called a k-morphism. A k-morphism has a groupoid structure
defined by identity, composition, and inverse operations. These operations satisfy the groupoid laws
which are associativity of composition, identity as a unit of composition and cancellation of inverses
through a weak sense of equality but only up to a morphism at the next level k+ 1. We can view the
k-morphism as a k-dimensional path in a topological space. Similarly, we can observe the elements of an
iterated identity type at level k as k-dimensional paths. Therefore a proof element of the type x =A y acts
like a one-dimensional path between endpoints x and y and a proof element of type m =x=Ay n acts like a
2-dimensional path or a homotopy between paths of type x =A y and so on. Moreover, these paths also
satisfy the groupoid laws up to homotopy at the next level in the following sense.

• re f l ◦ x = x◦ re f l = x −→ identity as a unit of composition

• (x◦ y)◦ z = x◦ (y◦ z) −→ associativity of composition

• !x◦ x = x◦ !x = re f l −→ cancellation of inverses

where re f l is an element of type x =A x.

Because of the correspondence of types to a topological space or a higher-dimensional groupoid,
we can map the elements of an identity type, which are paths in homotopy type theory, to equivalences
between types in a universe. Equivalence can be relaxed to a bijection when types behave like sets. The
mapping of a path to equivalence is made possible by the univalence axiom which describes that we may
identify equivalent types A and B in the following sense.

ua : (A' B)→ (A =U B) (1)

In (1), the type U is the universe or the type of types. The univalence axiom states that when we have
a proof of type A ' B, we can obtain a path between A and B. In homotopy type theory, the following

19



A Homotopical Approach to Cryptography P. Vivekanandan

defining equations give an equivalence between type A and type B.

A' B :≡ ∑
f :A→B

isequiv( f ) (2)

isequiv( f ) :≡
(

∑
g:B→A

( f ◦g∼ idB)
)
×
(

∑
h:B→A

(h◦ f ∼ idA)
)

(3)

A homotopy between non-dependent functions f1, f2 : A1→ A2 is given by the following equation.

f1 ∼ f2 :≡∏
x:A1

( f1(x) =A2 f2(x)) (4)

In (3), the composite f ◦ g is homotopic to the identity function idB, and the composite h ◦ f is
homotopic to the identity function idA. There is also a reduced notion of equivalence called quasi-inverse.
A quasi-inverse for a function f : A→ B is given by

qinv( f ) :≡ ∑
g:B→A

(
( f ◦g∼ idB)× (g◦ f ∼ idA)

)
(5)

Also, we have a function that maps an element of quasi-inverse qinv( f ) to isequiv( f ) for f : A→ B
[32].

mkqinv : qinv( f )→ isequiv( f ) (6)

For examples described in this paper, we will use mkqinv to obtain a proof of equivalence from quasi-
inverse. For a path p : A =U B, we have a function coe [5] that coerces along p. The following equation
gives the type of coe.

coe : (A =U B)→ (A→ B) (7)

In the presence of univalence, we also have a computation rule for coe [5] defined as follows.

coe(ua( f , isequiv( f )))x = f (x) (8)

where x : A, f : A→ B and ( f , isequiv( f )) : A' B.
Higher inductive types are a general schema for defining new types in homotopy type theory. It

extends an ordinary inductive type by providing constructors for generating paths and higher paths. In
homotopy type theory, we define a higher inductive type by specifying its introduction, elimination, and
computation rules. The introduction rule of a type specifies its constructors. The elimination rule of a
type defines how to use its elements, and the computation rule describes the action of the elimination
rule on the constructors of the type. A simple example for higher inductive type is the interval type
I. It consists of two point constructors 0I and 1I and a path constructor seg : 0I =I 1I . The following
declaration1 specifies the introduction rule for I.

1 data I : Set where

2

3 -- point constructors

4 zero : I

1In this paper, we have given a reduced declaration of higher inductive types for better understanding. In Agda, we use rewrite
rules to define higher inductive types where we specify the path constructors as postulates.

5 one : I

6

7 -- path constructors

8 seg : zero ≡ one

The non-dependent elimination rule or the recursion principle of I states that when given a type C
along with constructors c0,c1 : C and cseg : c0 =C c1, there is a function f :I→C such that f (zero) = c0,
f (one) = c1 and ap f (seg) = cseg where ap f defines the action of functions on paths. The equalities
f (zero) = c0, f (one) = c1 and ap f (seg) = cseg are the computation rules for the type I. The compu-
tational rules for the point constructors zero and one hold definitionally, but the computation rule for
path constructor seg holds only propositionally, and we specify it as an axiom which is a limitation of
homotopy type theory.

Similarly, the dependent eliminator or the induction principle of I states that when given a type
D :I→U along with constructors d0 : D(zero), d1 : D(one) and dseg : d0 =

D
seg d1, there is a dependent

function f : ∏(x:I) D(x) with computation rules f (zero) = d0, f (one) = d1 and apd f (seg) = dseg.
Here dseg is a heterogeneous path transported over seg and apd f defines the action of functions on
heterogenous paths [32].

Another important concept of homotopy type theory which is central to understand the idea proposed
in this paper is that the functions behave functorially on paths. It means that a function f : A→ B respects
equality and it preserves the path structure in the mapping from type A to type B. Now we can give the
type of ap f which defines the action of non-dependent functions on paths as follows.

ap f : (x =A y)→ ( f (x) =A f (y)) (9)

The following equation gives the action of dependent functions of type f : ∏(x:A) B(x) on paths.

apd f : ∏
p:x=y

(p∗( f (x)) =B(y) f (y)) (10)

In (10), p∗( f (x)) lying in space B(y) can be thought of as an endpoint of a path obtained by lifting p
from f(x) to a path in the total space ∑(x:A) B(x)→ A [32]. The following equation gives the type of p∗
also known as transport.

transportB
p : B(x)→ B(y) (11)

where p : x = y for x,y : A.
In the next section, we use the concepts discussed here to encode a cryptographic scheme. We define

the scheme using a higher inductive type and expose the implementation for mapping the higher inductive
type to a concrete implementation in the universe.

3 Higher Inductive Type front-end for OTP

In this section, we will discuss an encoding of the one-time pad using a higher inductive type with a
path constructor to specify the encryption function. We will construct a proof for an equivalence which
reflects the encryption path of the higher inductive type in the universe. The functional correctness prop-
erty, which states that decryption inverts encryption, will be part of the construction of the proof for
the equivalence. We will then map this higher inductive type, with the encryption path, to a concrete
implementation of the one-time pad, with the equivalence reflecting the encryption path, in the universe.
The encryption and the decryption functions are then projected from the concrete implementation in the

universe using the higher inductive type which acts as a front-end. By accessing the concrete implemen-
tation of the one-time pad through a higher inductive type, we can get a certificate or a guarantee on
the functional correctness of the system. Some other property such as homomorphic encryption requires
introducing higher-dimensional paths to act as a certificate. We will discuss higher-dimensional paths in
section 4.

3.1 One-time Pad

The following Agda code gives the higher inductive type encoding of the one-time pad.

1 data OTP (n : Nat) : Set where

2

3 -- point constructors

4 message : OTP n

5 cipher : OTP n

6

7 -- path constructors

8 otp-encrypt :

9 {n : Nat} ->

10 (key : Vec Bit n) ->

11 message {n} ≡ cipher {n}

The higher inductive type OTP has two point constructors message and cipher representing the
plain-text and the cipher-text respectively. The path constructor otp-encrypt represents the encryption
function of the one-time pad. We parameterize the type OTP with the length n of the data. otp-encrypt
uses the same length parameter n to specify the length of the key which encodes another restriction,
namely the length of the key for the one-time pad should be equal to the length of the message, which is
crucial for the security of the one-time pad.

The following code gives the recursion principle and its action on constructors or the computation
rules for the type OTP.

1 otp-rec :

2 {n : Nat} ->

3 (B : Set) ->

4 (b-msg : B) ->

5 (b-cipher : B) ->

6 (b-encrypt : (key : Vec Bit n) -> b-msg ≡ b-cipher) ->

7 OTP n -> B

8 otp-rec B b-msg b-cipher b-encrypt message = b-msg

9 otp-rec B b-msg b-cipher b-encrypt cipher = b-cipher

10

11 postulate

12 β-otp-rec :

13 {n : Nat} ->

14 (B : Set) ->

15 (b-msg : B) ->

16 (b-cipher : B) ->

17 (b-encrypt : (key : Vec Bit n) -> b-msg ≡ b-cipher) ->

18 {key : Vec Bit n} ->

19 ap (otp-rec B b-msg b-cipher b-encrypt)

20 (otp-encrypt key)

21 ≡
22 (b-encrypt key)

The recursion principle otp-rec states that when given a type B with point constructors b-msg and
b-cipher and path constructor b-encrypt, there exists a function of type OTP n -> B. otp-rec maps
message and cipher to b-msg and b-cipher respectively. β-otp-rec gives the action of otp-rec
on the path (otp-encrypt key) which maps it to the path (b-encrypt key). Equation (9) gives the
type of ap. The computation rules for point constructors message and cipher are given as definitional
equalities specified as part of otp-rec. The computation rule for the path otp-encrypt is postulated
as propositional equality.

The following code gives the induction principle and its computation rules for OTP.

1 otp-ind :

2 {n : Nat} ->

3 (B : OTP n -> Set) ->

4 (b-msg : B (message)) ->

5 (b-cipher : B (cipher)) ->

6 (b-encrypt : (key : Vec Bit n) ->

7 transport B (otp-encrypt key) b-msg ≡ b-cipher) ->

8 (x : OTP n) -> B x

9 otp-ind B b-msg b-cipher b-encrypt message = b-msg

10 otp-ind B b-msg b-cipher b-encrypt cipher = b-cipher

11

12 postulate

13 β-otp-ind :

14 {n : Nat} ->

15 (B : OTP n -> Set) ->

16 (b-msg : B (message)) ->

17 (b-cipher : B (cipher)) ->

18 (b-encrypt : (key : Vec Bit n) ->

19 transport B (otp-encrypt key) b-msg ≡ b-cipher) ->

20 {key : Vec Bit n} ->

21 apd (otp-ind B b-msg b-cipher b-encrypt)

22 (otp-encrypt key)

23 ≡
24 (b-encrypt key)

The induction rule otp-ind states that when given a type B : OTP n -> Set along with points b-msg,
b-cipher and path b-encrypt, there exists a dependent function (x : OTP n) -> B x. The com-
putation rule for path b-encrypt is postulated as propositional equality. Equation (10) gives the type of
apd and equation (11) gives the type of transport where p is the path (otp-encrypt key).

3.2 Implementation of one-time pad in the universe

The functional programming aspect of homotopy type theory allows us to implement any cryptographic
schemes. In this section, we will develop a concrete model for the higher inductive type OTP described
in section 3.1. The encryption function for the one-time pad is straightforward, and it is implemented
using xor. The encryption of one-time pad is defined using the following function.

1 OTP-encrypt :

2 {n : Nat} ->

3 (key : Vec Bit n) ->

4 (message : Vec Bit n) ->

5 Vec Bit n

6 OTP-encrypt {n} key message = message xorBits key

where xorBits perform xor on two vectors of equal length.

Similar to keys, we have chosen to use the type Vec Bit n to represent the point constructors message
and cipher of the higher inductive type OTP in the universe. Therefore, the path otp-encrypt should
be mapped to an equivalence formed by OTP-encrypt between types Vec Bit n and Vec Bit n. To
create an equivalence for the function OTP-encrypt, we need a proof element of type given by equation
(5). To construct a proof element of (5), we need a function g : Vec Bit n � Vec Bit n, a proof
element of f◦g ∼ id, and a proof element of g◦f ∼ id. For the one-time pad, the encryption function
is also its inverse. So both f and g are represented by OTP-encrypt in this case. Therefore, the types
f◦g ∼ id and g◦f ∼ id are definitionally the same. The equivalence formed by OTP-encrypt is
defined as follows.

1 OTP-equiv :

2 {n : Nat} ->

3 (key : Vec Bit n) ->

4 Vec Bit n ' Vec Bit n

5 OTP-equiv key = ((OTP-encrypt key) ,

6 equiv

7 (mkqinv

8 (OTP-encrypt key)

9 (α-OTP key)

10 (α-OTP key)))

11

12 (α-OTP key) : (OTP-encrypt key (OTP-encrypt key msg)) ≡ msg

In the above code, (OTP-equiv key) is of the type given by equation (2). equiv forms a proof
element of the type given by equation (3). The type of mkqinv is given by equation (6) which takes an
element of (5) as input and gives an element of (3) as output. (α-OTP key) is a proof element which
says that the encryption of msg, implemented by OTP-encrypt, followed by its decryption, which is
also implemented by OTP-encrypt in this case, is the same as msg. To form the proof element (α-OTP
key), we need to construct proofs for self-inverse, identity and associativity for bitwise application of
xor to a vector of bits.

1 xor-inv-pf :

2 {n : Nat} �

3 (x : Vec Bit n) �

4 x xorBits x

5 ≡
6 (0b-vec n)

7 xor-inv-pf =

8 (indVec

9 (λ {n} x � x xorBits x ≡ (0b-vec n))

10 refl

11 (λ {n} x xs pf �

12 begin

13 (x :: xs) xorBits (x :: xs)

14 ≡〈 refl 〉
15 (x xor x) :: (xs xorBits xs)

16 ≡〈 ap (λ pxs � (x xor x) :: pxs) pf 〉
17 (x xor x) :: (0b-vec n)

18 ≡〈 ap (λ px � px :: (0b-vec n)) (xor-inv x) 〉
19 0b :: (0b-vec n)

20 ≡〈 refl 〉
21 0b-vec (suc n) �))

xor-inv-pf says that when a vector x of length n is xor’ed with itself, it results in a 0-bit vector of
length n. We derive the proof for this property using the induction principle of vec type.

1 xor-id-pf :

2 {n : Nat} �

3 (x : Vec Bit n) �

4 x xorBits (0b-vec n)

5 ≡
6 x

7 xor-id-pf =

8 indVec

9 (λ {n} x � x xorBits (0b-vec n) ≡ x)

10 refl

11 λ {n} x xs pf �

12 begin

13 (x :: xs) xorBits 0b-vec (suc n)

14 ≡〈 refl 〉
15 (x :: xs) xorBits (0b :: (0b-vec n))

16 ≡〈 refl 〉
17 (x xor 0b) :: (xs xorBits (0b-vec n))

18 ≡〈 ap (λ pxs � (x xor 0b) :: pxs) pf 〉
19 (x xor 0b) :: xs

20 ≡〈 ap (λ px � px :: xs) (xor-id x) 〉
21 (x :: xs) �

xor-id-pf says that when a vector x of length n is xor’ed with a 0-bit vector of length n, it results
in the same vector x of length n. Again, we use the induction principle of vec to derive the proof for
xor-id-pf.

1 xor-assoc-pf :

2 {n : Nat} �

3 (x : Vec Bit n) �

4 (y : Vec Bit n) �

5 (z : Vec Bit n) �

6 (x xorBits y) xorBits z

7 ≡
8 (x xorBits (y xorBits z))

9 xor-assoc-pf {n} [] [] [] = refl

10 xor-assoc-pf {n} (x :: xs) (y :: ys) (z :: zs) =

11 begin

12 ((x :: xs) xorBits (y :: ys)) xorBits (z :: zs)

13 ≡〈 refl 〉
14 ((x xor y) :: (xs xorBits ys)) xorBits (z :: zs)

15 ≡〈 refl 〉
16 ((x xor y) xor z) :: ((xs xorBits ys) xorBits zs)

17 ≡〈 ap (λ p � p :: ((xs xorBits ys) xorBits zs)) (xor-assoc x y z) 〉
18 (x xor (y xor z)) :: ((xs xorBits ys) xorBits zs)

19 ≡〈 ap (λ p � (x xor (y xor z)) :: p) (xor-assoc-pf xs ys zs) 〉
20 (x xor (y xor z)) :: (xs xorBits (ys xorBits zs))

21 ≡〈 refl 〉
22 (x :: xs) xorBits ((y xor z) :: (ys xorBits zs))

23 ≡〈 refl 〉
24 (x :: xs) xorBits ((y :: ys) xorBits (z :: zs)) �

xor-assoc-pf is the proof element describing the associativity of xor operation between three vec-
tors x, y and z of length n.

1 α-OTP :

2 {n : Nat} �

3 (key : Vec Bit n) �

4 (message : Vec Bit n) �

5 (OTP-encrypt {n} key (OTP-encrypt {n} key message))

6 ≡
7 message

8 α-OTP {n} key message =

9 begin

10 (OTP-encrypt key (OTP-encrypt key message))

11 ≡〈 refl 〉
12 (OTP-encrypt key (message xorBits key))

13 ≡〈 refl 〉
14 ((message xorBits key) xorBits key)

15 ≡〈 xor-assoc-pf message key key 〉
16 (message xorBits (key xorBits key))

17 ≡〈 ap (λ x � message xorBits x) (xor-inv-pf {n} key) 〉
18 (message xorBits (0b-vec n))

19 ≡〈 xor-id-pf {n} message 〉
20 message �

Finally, we derive the proof element for α-OTP using the proof elements of xor-id-pf, xor-inv-pf
and xor-assoc-pf.

3.3 Mapping OTP into the universe

The higher inductive type OTP defined in section 3.1 can now be mapped into the universe using univa-
lence. The abstract nature of higher inductive types also means that we can map the same type to more
than one concrete implementation in the universe whenever compatible. The equivalence (OTP-equiv

key) respects the path structure specified by the constructor otp-encrypt. Because of this, a path
formed by univalence given by (ua (OTP-equiv key)) represents the path structure of otp-encrypt
in the universe. This correspondence allows us to define a mapping I-OTP which maps the points
message, cipher of OTP to type Vec Bit n and a mapping I-OTP-pathwhich maps the path (otp-encrypt
key) to (ua (OTP-equiv key)).

1 I-OTP : {n : Nat} -> OTP n -> Set

2 I-OTP {n} bits = otp-rec Set

3 (Vec Bit n)

4 (Vec Bit n)

5 (λ key -> ua (OTP-equiv key))

6 bits

7

8 I-OTP-path : {n : Nat} ->

9 (key : Vec Bit n) ->

10 ap I-OTP (otp-encrypt {n} key) ≡ ua (OTP-equiv key)

11 I-OTP-path {n} key = β-otp-rec Set

12 (Vec Bit n)

13 (Vec Bit n)

14 (λ k -> ua (OTP-equiv k))

I-OTP is defined using the recursion principle otp-rec of the higher inductive type OTP. It maps the
points of OTP to the type Vec Bit n in the universe represented by Set. I-OTP-path maps the path
(otp-encrypt key) to (ua (OTP-equiv key)) using β-otp-rec. Now we can define an interpreter
function ITP using coe given by equation (7) as follows.

1 ITP : {n : Nat} ->

2 {a b : OTP n} ->

3 (p : a ≡ b) ->

4 (I-OTP a) ->

5 (I-OTP b)

6 ITP {n} {a} {b} p = coe (ap I-OTP p)

When we give the path otp-encrypt as input, the interpreter ITP returns the encryption function
OTP-encrypt. By accessing a concrete implementation in the universe using a higher inductive type, we
get the certificate or guarantee specified by the path structures of the higher inductive type. In the case of
OTP, the functional correctness property is part of the equivalence (OTP-equiv key) given by (α-OTP

key), and the path otp-encrypt will reflect this through the mapping specified by I-OTP-path.
We will consider an example of using ITP to extract OTP-encrypt and its application on a vector.

pf : (ITP

(otp-encrypt

(1b :: (0b :: [])))

(1b :: (1b :: [])))

≡
(0b :: (1b :: []))

In the above code, ITP takes otp-encrypt as input with key (1b :: (0b :: [])) and plain-
text (1b :: (1b :: [])) and returns the cipher-text (0b :: (1b ::[])) as output.

4 Encoding Properties as Higher Dimensional Paths

The path otp-encrypt described in the previous section is one-dimensional. We can also encode
domain-specific cryptographic properties as higher dimensional paths. In this section, we will design
properties of a database model with multi-layered encryptions in the style of cryptDB [28] as higher
dimensional paths. CryptDB has different layers of encryption known as onion layers of encryption. The
idea of cryptDB is to allow computation on top of encrypted data without the need to decrypt them. For
example, homomorphic encryption can be used to implement addition, and deterministic encryption can
be used to perform equality comparison on top of encrypted data. Similarly, order-preserving encryption
can be used to implement inequality comparisons on encrypted data. A higher inductive type can be
used to define the computational behavior of cryptDB. We will consider the following higher inductive
type specification to discuss encoding domain-specific laws of cryptDB as higher dimensional paths2.
CryptDB involves non-bijective functions, and can be implemented using singleton types [5]. In this
section, we will not be focusing on the implementation details or mapping types into the universe.

1 data encDB : Set where

2

3 -- point constructors

4 tab : encDB

5 tabDET : encDB

6 tabHOM : encDB

7 tabOPE : encDB

8

9 -- one-dimensional paths

10 hom-enc : tab ≡ tabHOM

11 det-enc : tab ≡ tabDET

12 ope-enc : tab ≡ tabOPE

2We have simplified the higher inductive type encDB for ease of understanding. A detailed implementation can be found in the
associated github repository.

tabHOM tabHOM

tabtab

encincrement

increment

 paillierhomhomenc ! homenc

Figure 1: Homotopy representing the homomorphic property of paillier cryptosystem. The path hom-enc
concatenated with enc-increment and (! hom-enc) is equal to the path increment.

The higher inductive type encDB specifies a lot of restrictions and a mapping to a concrete implemen-
tation should respect those restrictions. For example, it says that homomorphic encryption is a function
that takes a plain-text table tab as input and gives an encrypted version of the table tabHOM as output.
The inverse path (! hom-enc) specifies the decryption function. Similarly, the paths det-enc and
ope-enc specifies the deterministic and order-preserving encryption schemes respectively. The higher
inductive type encDB acts as a single interface giving a lot of information on underlying implementation
of a cryptographic setting. It provides us with a graphical model composed of points, paths, paths be-
tween paths or higher dimensional paths to specify the correctness properties and various domain-specific
laws of a cryptographic construction. In the remainder of this section, we will discuss homotopies or path
between paths describing properties specific to homomorphic encryption, deterministic encryption, and
order-preserving encryption.

4.1 Homomorphic Encryption

Homomorphic encryption can be used to perform computations on cipher-text. In cryptDB, homomor-
phic encryption is implemented using paillier cryptosystem. According to the homomorphic property
of paillier cryptosystem [26], the addition of two plain-texts will be equal to the multiplication of their
corresponding cipher-text. We can express this property as a two-dimensional path saying homomorphic
encryption of a plain-text concatenated with a path expressing homomorphic multiplication concatenated
with homomorphic decryption is the same as the regular addition performed on the plain-text.

The encoding of cryptDB in homotopy type theory involves non-bijective queries. Mapping a non-
bijective query into the universe is not possible in the current type-theoretic setting. However, we can
map a non-bijective path to singleton types in the universe [5]. Such a mapping holds because any
function between singleton types is automatically a bijection.

4.2 Deterministic Encryption

Deterministic encryption generates the same cipher-text on multiple encryptions of the same plain-text.
In cryptDB, a deterministic encryption scheme is used to perform equality comparisons on encrypted
data. The correctness property of deterministic encryption requires DET (m1) ≡ DET (m2) when m1 ≡
m2. We can specify this property as a heterogenous path over a path of type m1 ≡ m2. For example,

m1 m2
p

f(m2)f(m1)

f(p)

A

∑(x : A) 
P(x)

pr1

Figure 2: In the above figure [32], the function f represents det-enc. The path f (p) lies in the total
space over the path p.

when tab and det-enc encode the plain-text as an implicit argument given by tab : {m} � encDB

and det-enc : {m} � tab ≡ tabDET respectively, we can define the following two-dimensional
path.

det-correctness :

(p : m1 ≡ m2) �

transport (λ x � tab {x} ≡ tabDET) p (det-enc {m1})

≡
(det-enc {m2})

det-correctness says that the path (det-enc {m1}) ≡ (det-enc {m2}) lies over p : m1

≡ m2.

4.3 Order-Preserving Encryption

Order-preserving encryption [2] allows inequality comparisons on encrypted data without the need to
decrypt them. Order-preserving encryption requires, for plain-texts x and y, if (x < y) then OPE(x) <
OPE(y). We cannot specify this property in the style of det-correctness because inequality relation
does not form paths. However, we can use a different approach to model this restriction in a higher
inductive type. For example, consider a function bigE (m1, m2) which returns the biggest of two
elements. When there exists a path p’ : bigE(m1, m2) ≡ bigE(c1, c2), where c1 and c2 are the
OPE cipher values of m1 and m2 respectively, lying in the space encDB, we can design a two-dimensional
path saying ope-encrypt is the same path as p’. This two-dimensional path will hold only when the
order-preserving encryption respects the inequality relation between the plain-texts.

The two-dimensional paths discussed above capture different domain-specific laws that should be
respected by any concrete implementation of a multi-layered database in the style of cryptDB. By spec-
ifying the above paths as constructors of encDB and by mapping encDB to a concrete implementation

in the universe similar to OTP in section 3, we can achieve various guarantees on the correctness of the
implementation. The mapping of the higher inductive type into the universe alone is enough to guarantee
on the correctness of properties specified by the path constructors because of univalence and functorial-
ity. By having a higher inductive type front-end for a cryptographic implementation, we eliminate the
need to generate individual proofs for different domain-specific properties. Also in a higher inductive
type framework, we have a way to relate proofs of different properties because of the encoding of proofs
as paths or higher dimensional paths of a single type.

5 Implementation

In cryptDB, functions implementing queries like insert and delete are not bijective and therefore can-
not be encoded as paths in a higher inductive type. To address this problem, we used the patch theory
[5] approach to encode cryptDB. The higher inductive type representing cryptDB is contractible which
allowed us to map the paths representing the non-bijective queries to singleton types in the universe.
This mapping is possible since any function between a singleton type is automatically a bijection. We
declared the query operations using historytype [5], a higher inductive type that records all the query
information. The higher inductive type representing the cryptDB then depends on the history type.
We implemented this model in Agda3 by declaring the higher inductive types as rewrite rules [13] us-
ing {-# REWRITE , ...#-} pragma. We automated the code generation for the dependent and non-
dependent elimination rules corresponding to the higher inductive types using an automation tool [33]
based on Agda’s new support for elaborator reflection [12].

6 Related Work

The work discussed in this paper takes the first step towards formal specification of cryptographic proto-
cols based on types. There are other works which support formal specification of cryptographic construc-
tions using different settings for handling cryptographic primitives including shared-key and public-key
cryptography, signatures, hash functions, message authentication codes, etc. In this section, we will
review few of those works.

6.1 Foundational Cryptography Framework

The Foundational Cryptography Framework (FCF) [27] implements a probabilistic programming lan-
guage embedded inside Coq proof assistant. Unlike Agda, the Coq proof assistant is based on the
Calculuso f InductiveConstruction. However, the recent version of Coq allows the sort Set to be pred-
icative. The probabilistic programming language defined by FCF enables the specification of crypto-
graphic schemes, security definitions, and hard problems. A shallow embedding of the probabilistic
language allows FCF to have access to the capabilities of the metalanguage (Coq) including dependent
types and higher-order functions. It also allows any theory developed in the host language accessible to
the embedded language. The technique described in this paper uses a higher inductive type to specify
cryptographic protocols in a formal setting. Coq does not have a built-in mechanism to support the def-
inition of higher inductive types. However, we can still work with higher inductive types in Coq. With
further work, the shallow embedding can make the constructions involving higher inductive types visible
to the Foundational Cryptography Framework.

3A detailed implementation can be found in the associated github repository. See https://github.com/pavenvivek/JISIS-2018.

6.2 CryptoVerif

The work of [10] implemented in CryptoVerif provides a mechanized prover for showing correspon-
dence assertions which are useful to express authentication properties for cryptographic protocols in the
computational model. The proof construction follows the sequences of games approach in cryptography.
CryptoVerif is based on ProcessCalculus extended with parametric events to serve in the definition of
correspondences. The work also discusses proving mutual authentication and authenticated key exchange
using correspondences. CryptoVerif incorporates efficient automation reducing the proof development
effort but lacks interactive proof development features which makes it more specific to only a subset of
cryptographic constructions when compared to FCF or EasyCrypt.

6.3 ProVerif

ProVerif [11] is a cryptographic protocol verifier for the automated reasoning of security properties based
on Dolev-Yao model. It can be used for proving secrecy, authentication, and equivalences between
processes differing only by terms. The input protocols to ProVerif are modeled using PiCalculus and
internally translated using Horn clauses. The security properties which needs to be proved are translated
to derivability queries on these clauses. ProVerif can handle different cryptographic primitives including
shared-key and public-key cryptography, hash functions, and Diffie-Hellman key agreements.

6.4 EasyCrypt

CertiCrypt [6], a framework built upon the Coq proof assistant, enables machine-checked construction
and verification of cryptographic schemes. The proof development in CertiCrypt is time-consuming,
and EasyCrypt [7] was developed to address this limitation by speeding up the construction of proofs
using automation based on SMT solvers. Both CertiCrypt and EasyCrypt has a deep embedding of a
probabilistic programming language which is used for proof construction. The deep embedding makes
them inaccessible to the cozy features of the host language (Coq) such as dependent-types, higher-order
functions, modules, etc.

6.5 Verypto

Verypto [9], a framework implemented in Isabelle proof-assistant [24], provides a formal language for
the specification and verification of game-based cryptographic security proofs. Verypto includes a prob-
abilistic higher-order functional programming language with recursive types, references, and events to
express constructs of a game-based security proof. The language handles stateful higher-order objects
such as oracles, arbitrary data types and supports event-based reasoning patterns. Like CertiCrypt and
EasyCrypt, the probabilistic programming language used for proof construction in Verypto follows a
deep embedding.

7 Limitations and Future Work

A limitation of homotopy type theory is that the univalence can be added only as an axiom. This limita-
tion weakens the good computational properties of type theory. We would like to develop the framework
described in this paper using cubical type theory [14]. In cubical type theory, the univalence computes
and is no longer an axiom.

Another limitation is that the mapping of higher inductive type into the universe requires the functions
represented by paths to be bijective. We cannot specify all functions as bijections. In the case of cryptDB,

functions implementing queries like insert and delete are not bijective and therefore cannot be encoded as
paths in a higher inductive type. The functions with simple retractions are not acceptable. Every function
should have inverses to be expressed as paths. One way to work around this problem is to encode
functions as mappings between singleton types in the universe [5]. Any function mapping between
two singleton types is automatically a bijection. So a path representing a non-bijective function in a
higher inductive type can be mapped to bijection formed by a function between singleton types in the
universe. Future work in this direction would be to characterize mapping of partial bijections to paths
using the tools of homotopy theory. Another direction is to develop type theory with non-symmetric
paths based on directed type theory [21]. In the current setting, since homotopy type theory is also a
functional programming language, the non-bijective functions can be used along with higher inductive
types. So the benefits of having a higher inductive type representing bijective functions can still be
achieved. Homotopy type theory also allows us to postulate bijections as axioms and work with them.
When we have a proof that a function is bijective in a different setting, then the function can be postulated
as a bijection in homotopy type theory and can be encoded using a higher inductive type.

Probabilistic encryption schemes are not bijective. It might not be possible to map them to singleton
types in the universe because they compute to different values during each execution with overwhelming
probability and does not uniquely identify the contents of a singleton type. So the probabilistic encryption
schemes have to be encoded as regular mappings and can be used along with higher inductive types.
Another limitation is the difficulty involved in deriving proofs for bijections. This limitation increases
development time and effort. But after application development, we can achieve overwhelming guarantee
on the correctness of the application. In the real-world applications, bug fixing has taken much more
effort than the original development effort [8][18][15]. So the cost of the increase in development effort
can be ignored considering the benefits achieved. It can be very significant especially when implementing
cryptographic protocols because a flawed implementation of cryptographic protocol leads to serious
security issues resulting in the compromise of the entire application. Agda also has a robust reflection
library which can be used to automate the generation of proofs [19] and elimination rules for higher
inductive types. Automated code generation can reduce the development effort to some extent. In the
future, we would like to encode the security properties of a cryptographic scheme as paths in a higher
inductive type and explore how to achieve security guarantees using this setting.

The main purpose of the discussion in this paper is to drive the progress in homotopy type theory
research from an application perspective on the programming side. Cryptography is a very significant
and vast field, and it would be interesting to see if homotopy type theory can find application in this
domain. This paper takes a very first step towards this direction.

8 Conclusion

This paper presented a new direction for the formal specification of cryptographic protocols based on
types. It gave a real-world application of homotopy type theory in an attempt to solve an important
problem in cryptography, namely verifying the correctness of the implementation. It also extended the
types in an interesting way by allowing them to act as formal certificates guaranteeing on the correctness
properties. Homotopy type theory is still developing and it takes more time and more hard work to get
it done. In the meantime, the current features of homotopy type theory such as the higher inductive type
and the univalence axiom have been put to use by this paper to model an industrial application. Applying
homotopy type theory to cryptography is an important topic to explore, and this paper can motivate more
research in this direction. In spite of the limited scope of this framework, we still feel this discussion is
necessary because the ongoing works [4][14] are promising, and it can motivate more research on the
programming side of homotopy type theory.

The limitations of homotopy type theory, namely having univalence only as an axiom and the re-
quirement for functions to have inverses has restricted us to only a subset of cryptographic schemes to be
benefitted by the model described in this paper. Nevertheless, there is a lot of work going on to improve
type theory to allow for univalence to compute and mapping of non-bijective functions into the universe
which can reduce the restrictions and enable us to encode more interesting cryptographic constructions
using the higher inductive type model. Also, this paper introduces the tools of homotopy type theory
to the cryptographic community and acts as a precursor of more interesting type theoretical settings to
follow which can significantly improve the framework described in this paper.

